


Ausgleichsvorgänge in elektro-mechanischen Systemen mit Maple analysieren

Grundwissen für Antriebstechnik und Mechatronik

2011 | 1. Auflage

Modellierung und Analyse von Stromrichterschaltungen

Gleichrichter in Zweipuls-Mittelpunktschaltung



Das Bild zeigt die Schaltung des Gleichrichters (M2-Schaltung) mit ohmsch-induktiver Last (R_3 , L_3). Die Induktivitäten L_1 und L_2 berücksichtigen zusammenfassend die Streureaktanzen des Stromrichtertransformators, die Reaktanzen der Kommutierungs-drosseln und des speisenden Netzes. Die ohmschen Widerstände der Transformatorwicklungen, der Drosseln und der Leitungen repräsentieren die Widerstände R_1 und R_2 . In Abhängigkeit vom Schaltzustand der beiden Dioden kann man drei Fälle unterscheiden:

| | Diode 1 | Diode 2 |
|---------|----------|----------|
| Fall 1: | leitend | sperrend |
| Fall 2: | sperrend | leitend |
| Fall 3: | leitend | leitend |

Fall 3 ist der Kommutierungszustand. Unter dem Einfluss von Netz- und Lastspannung geht eine der Dioden in den Zustand "leitend" über, während die andere noch Strom führt. Zwischen diesen Zweigen fließt daher ein Kommutierungsstrom, der den Strom in einem der Zweige ab- und im anderen Zweig aufbaut.

Die Dioden werden im vorliegenden Beispiel als ideale Schalter betrachtet, d. h. sie führen im Sperrzustand keinen Strom. Beim Aufstellen des Modells für die genannten Fälle kann man daher den Zweig mit einer sperrenden Diode als nicht existent ansehen. Somit wird jeder Fall durch ein separates Gleichungssystem beschrieben. Nach dieser Vorgehensweise erhält man mit dem Lastzweig als vollständigen Baum [1, Kapitel 4] folgende Maschengleichungen:

$$\text{Fall 1: } M1 := (L1 + L3) \left(\frac{d}{dt} i1(t) \right) + (R1 + R3) i1(t) = u1(t)$$

$$\text{Fall 2: } M2 := (L2 + L3) \left(\frac{d}{dt} i2(t) \right) + (R2 + R3) i2(t) = u2(t)$$

Fall 3:

$$M3_1 := (L1 + L3) \left(\frac{d}{dt} i1(t) \right) + (R1 + R3) i1(t) + R3 i2(t) + L3 \left(\frac{d}{dt} i2(t) \right) = u1(t)$$

$$M3_2 := (L2 + L3) \left(\frac{d}{dt} i2(t) \right) + (R1 + R3) i2(t) + R3 i1(t) + L3 \left(\frac{d}{dt} i1(t) \right) = u2(t)$$

Die Maschengleichungen M1 und M2 liefern durch Umstellung die Differentiale di_1/dt für den Fall 1 und di_2/dt für den Fall 2. Die beiden Maschengleichungen zum Fall 3 haben weitgehende Ähnlichkeit mit denen des Beispiels "Drehstromdrossel" im Abschnitt 3.4.1 von [1]. In jeder Gleichung erscheinen die Ableitungen der beiden Ströme i_1 und i_2 . Durch einfaches Umstellen der Gleichungen nach einer der Ableitungen ist keine Differentialgleichung zu gewinnen, für deren rechte Seite alle Anfangswerte bekannt sind (algebraische Schleife). Dieses Problem lässt sich jedoch wie in [1] durch Auflösung des linearen Gleichungssystems nach den beiden ersten Ableitungen von i_1 und i_2 beheben.

Maple-Programm:

```
> restart;
> M1:= (L1 + L3)*diff(i1(t),t) + (R1 + R3)*i1(t) = u1(t);
      M1 := (L1 + L3)  $\left( \frac{d}{dt} i1(t) \right)$  + (R1 + R3) i1(t) = u1(t)
> M2:= (L2 + L3)*diff(i2(t),t) + (R2 + R3)*i2(t) = u2(t);
      M2 := (L2 + L3)  $\left( \frac{d}{dt} i2(t) \right)$  + (R2 + R3) i2(t) = u2(t)
> M3_1:= (L1+L3)*diff(i1(t),t)+(R1+ R3)*i1(t) +
      R3*i2(t)+L3*diff(i2(t),t) = u1(t);
      M3_1 := (L1 + L3)  $\left( \frac{d}{dt} i1(t) \right)$  + (R1 + R3) i1(t) + R3 i2(t) + L3  $\left( \frac{d}{dt} i2(t) \right)$  = u1(t)
> M3_2:= (L2+L3)*diff(i2(t),t)+(R1+ R3)*i2(t) +
      R3*i1(t)+L3*diff(i1(t),t) = u2(t);
      M3_2 := (L2 + L3)  $\left( \frac{d}{dt} i2(t) \right)$  + (R1 + R3) i2(t) + R3 i1(t) + L3  $\left( \frac{d}{dt} i1(t) \right)$  = u2(t)
```

Umstellung der Maschengleichungen M1 und M2:

```
> G1:= isolate(M1, diff(i1(t),t));
      G1 :=  $\frac{d}{dt} i1(t) = \frac{u1(t) - (R1 + R3) i1(t)}{L1 + L3}$ 
> G2:= isolate(M2, diff(i2(t),t));
      G2 :=  $\frac{d}{dt} i2(t) = \frac{u2(t) - (R2 + R3) i2(t)}{L2 + L3}$ 
```

Auflösung der durch M3_1 und M3_2 gebildeten algebraischen Schleife:

```
> G3:= solve({M3_1, M3_2}, [diff(i1(t),t), diff(i2(t),t)]);
```

$$G3 := \left[\left[\frac{d}{dt} i1(t) = -\frac{1}{L1 L2 + L1 L3 + L3 L2} (L2 i1(t) R1 + i1(t) R3 L2 + R3 i2(t) L2 - L2 u1(t) + i1(t) R1 L3 - L3 i2(t) R1 - u1(t) L3 + L3 u2(t)), \frac{d}{dt} i2(t) = -\frac{1}{L1 L2 + L1 L3 + L3 L2} (-i1(t) R1 L3 + u1(t) L3 + L1 i1(t) R3 + L1 i2(t) R1 + L1 R3 i2(t) - L1 u2(t) + L3 i2(t) R1 - L3 u2(t)) \right] \right]$$

```
> collect(G3, [i1(t), i2(t)]);
```

$$\left[\left[\frac{d}{dt} i1(t) = -\frac{(L2 R1 + R3 L2 + R1 L3) i1(t)}{L1 L2 + L1 L3 + L3 L2} - \frac{(R3 L2 - R1 L3) i2(t)}{L1 L2 + L1 L3 + L3 L2} - \frac{L3 u2(t) - L2 u1(t) - u1(t) L3}{L1 L2 + L1 L3 + L3 L2}, \frac{d}{dt} i2(t) = -\frac{(-R1 L3 + L1 R3) i1(t)}{L1 L2 + L1 L3 + L3 L2} - \frac{(L1 R1 + L1 R3 + R1 L3) i2(t)}{L1 L2 + L1 L3 + L3 L2} - \frac{u1(t) L3 - L3 u2(t) - L1 u2(t)}{L1 L2 + L1 L3 + L3 L2} \right] \right]$$

Die Spannung u_L über dem Lastzweig wird beschrieben durch

```
> G4:= uL(t) = R3*(i1(t)+i2(t)) +
      L3*(diff(i1(t),t)+diff(i2(t),t));
G4 := uL(t) = R3 (i1(t) + i2(t)) + L3 \left( \frac{d}{dt} i1(t) + \frac{d}{dt} i2(t) \right)
```

Weil jede der Gleichungen nur unter bestimmten Bedingungen gültig ist, müssen noch geeignete Schaltbedingungen formuliert werden. Dazu werden die logischen Größen D1 und D2 eingeführt. D1 hat den Wert "wahr" (true), wenn Diode 1 leitend ist, also im Fall 1, sonst den Wert "falsch". Sinngemäß gilt das für D2 mit Diode 2. Leitend ist eine Diode, wenn sie Strom führt ($i_V > 0$) oder wenn nach dem Sperrzustand wieder die Bedingung für das Einsetzen des Stromflusses erfüllt ist ($u_1 > u_L$ bzw. $u_2 > u_L$).

Im Programm wird jedoch wie in [2] statt u_L die geringfügig verzögerte Spannung u_{Lv} zur Bildung von D1 und D2 verwendet, um algebraische Schleifen über i_1 , i_2 und u_L zu vermeiden. u_{Lv} wird durch ein Verzögerungsglied 1. Ordnung gebildet.

```
> DG1:= diff(uLv(t),t) = (uL(t)-uLv(t))/0.00001;
```

$$DG1 := \frac{d}{dt} uLv(t) = 1.10^5 uL(t) - 1.10^5 uLv(t)$$

```
> D1:= (i1(t)>0) or (u1(t) > uLv(t));
```

$$D1 := 0 < i1(t) \text{ or } u1(t) < uLv(t)$$

```
> D2:= (i2(t)>0) or (u2(t) > uLv(t));
```

$$D2 := 0 < i2(t) \text{ or } u2(t) < uLv(t)$$

Mit Hilfe des Befehls **piecewise** werden nun die Differentialgleichungen für di_1/dt und di_2/dt stückweise aus den oben für die einzelnen Fälle ermittelten Gleichungen zusammengesetzt. Dabei ist darauf zu achten, dass die Auswertung der Bedingungen von links nach rechts erfolgt und nach der ersten gültigen Bedingung beendet wird.

```
> DG2:= diff(i1(t),t) = piecewise(D1 and D2, rhs(G3[1,1]),
                                D1 and (not(D2)), rhs(G1), 0):
> DG3:= diff(i2(t),t) = piecewise(D1 and D2, rhs(G3[1,2]),
                                D2 and (not(D1)), rhs(G2), 0):
```


Vor der Lösung des Gleichungssystems mit **dsolve** sind noch die Gleichungen für die Netzspannungen u_1 und u_2 zu notieren und die Werte der Parameter festzulegen.

```
> G5:= u1(t) = sqrt(2)*Ueff*sin(omega*t);
                                G5:=u1(t)=sqrt(2) Ueffsin(omega*t)
> G6:= u2(t) = -sqrt(2)*Ueff*sin(omega*t);
                                G6:=u2(t)=-sqrt(2) Ueffsin(omega*t)
> Ueff:=100: omega:=314: R1:=2: R2:=2: R3:=10: L1:=0.04:
  L2:=0.04: L3:=0.2:
> Loe:= dsolve({DG1,DG2,DG3,G4,G5,G6,i1(0)=0,i2(0)=0,uLv(0)=0},
              numeric, [uL(t), i1(t),i2(t),u1(t),u2(t),uLv(t)],
              output=listprocedure, interr=false, maxfun=0);
Loe:= [t=proc(t) ... end proc, uL(t)=proc(t) ... end proc, i1(t)=proc(t) ... end proc,
       i2(t)=proc(t) ... end proc, u1(t)=proc(t) ... end proc, u2(t)=proc(t)
       ...
       end proc, uLv(t)=proc(t) ... end proc]
```


Die im Befehl **dsolve** verwendete Option **interr = false** wird im Abschnitt 3.4.5 von [1] beschrieben. Ohne diese Angabe kommt es zu einem vorzeitigen Abbruch der numerischen Rechnung, weil die Zwischenergebnisse eine Singularität vortäuschen.

Die Teillösungen werden nun Variablen zugewiesen und anschließend graphisch dargestellt.

```
> i1a:= subs(Loe, i1(t)):
> i2a:= subs(Loe, i2(t)):
> uLa:= subs(Loe, uL(t)):
> plot([i1a,i2a], 0..0.06, labels=["t/s", "i1/A, i2/A"],
       legend=["i1(t)", "i2(t)"]);
```



```
> with(plots):
> p1:= plot(i1a+i2a, 0..0.06, color=blue, labels=["t/s",
    "iL/A"], legend=["iL(t)"]):
> p2:= plot(uLa, 0..0.06, color=black, labels=["t/s", "uL/V"],
    legend=["uL(t)"]):
> dualaxisplot(p1,p2);
```



Literatur:

- [1] Müller, R.: Ausgleichsvorgänge in elektro-mechanischen Systemen mit Maple analysieren. Vieweg+Teubner Verlag 2011
- [2] Jentsch, W.: Digitale Simulation kontinuierlicher Systeme. R. Oldenbourg Verlag. München und Wien 1969

Ausgleichsvorgänge in elektro-mechanischen Systemen
mit Maple analysieren

Grundwissen für Antriebstechnik und Mechatronik

Müller, R.

2011, XII, 284 S. 69 Abb. Mit zahlreichen Beispielen und

Maple-Plots., Softcover

ISBN: 978-3-8348-1217-9